
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
% 
%                                                                          
%
%    File:           ch7_russian_deriv2.dtr                                 
%
%    Purpose:        Demonstrate basic idea of derivation through           
%
%                    inheritance                                            
%
%    Author:         Andrew Hippisley 20.07.10                          
%
%    Email:          andrew.hippisley@uky.edu                               
%
%    Address:        University of Kentucky, Lexington KY 40506-0027        
%
%    Documentation:  Chapter 7 'Derivation'                             
%
%                                                    
%
%    Related Files:  ch7_rusnoms.dtr, rusn9dec.dtr, rusa9dec.dtr            
%
%    Version:        1.00                                                  
%
%                                                                        
%
%                                                                           
%
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%
%                                                                           
%
%    Comment: Relates to Chapter 7, section 7.2.                            
%
%             Introduces inheritance-based derivational relatedness.        
%    
%             Default-based account of (1) category changing derivation,    
%
%             (2) conversion, (3) transposition and (4) category preserving 
%
%             derivation.  For (4) distinguishes head marking from non-     
%
%             head marking type                                             
%
%                                                                           



% 
%           
% 
%                                                                           
%
%            
%
%                                                                           
%
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%
%         
%
%    1. LOAD RUSSIAN INFLECTIONAL THEORY, SHOW PATH FILES FOR NOUNS/ADJS    
%
%                                                           
%
%              
%
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%

#load 'ch7_rusnoms.dtr'.
#load 'rusn9dec.dtr'.    %comment out ' rusn9dec.dtr' or 'rusa9dec.dtr'
%#load 'rusa9dec.dtr'.
#load 'hide_deriv.dtr'.

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%
%         
%
%    2 MORPHOTACTIC  GENERALIZATION NODES (see page 260)                    
%
%                                                                           
%
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%

SUFFIXATION:



    <stem> == "<base stem 1>" "<deriv affix>".

PREFIXATION:
   <stem> == "<deriv affix>" "<base stem>".

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%
%         
%
%    3 PRODUCTIVE LEXEME FORMATION TEMPLATES                                
%                                                                    %

                                      %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%
%         
%
%    3.1 CATEGORY CHANGING DERIVATION (page 260)                            
%                                                                    %

                                      %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%

LFT_PERSON:
    <> == NOUN
    <gloss> == λx [ <sem feature> (x) & "<base gloss>" (x) ]
    <sem feature> == person
    <stem> == SUFFIXATION   
    <deriv affix> == tel'.

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%
%         
%
%    3.2 CONVERSION (pages 262-63)                                          
%                                                                    %

                                      %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%

LFT_QUAL_ADJ:



    <> == ADJ
    <gloss> == λx [ <sem feature> (x, "<base gloss>") ]
    <sem feature> == possess_quality_of.

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%
%         
%
%    3.3 TRANSPOSITION (pages 264-65)                                       
%                                                                    %

                                      %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%

LFT_NOMINALIZATION:
    <> == NOUN
    <syn case_assign> == "<base syn case_assign>" %for e.g. Upravlenie
    <gloss> == "<base syn sem feature>" "<base gloss>" <syn sem feature>
    <stem> == SUFFIXATION
    <deriv affix> == enij
    <declensional_class> == N_IV:<mor>.

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%
%         
%
%    3.4 CATEGORY PRESERVING DERIVATION (pages 265-269)                     
%                                                                    %

                                      %
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%

LFT_CAT_PRESERVING:
    <> == LEXEME
    <syn> == "<base syn>"
    <gloss> == "<sem feature>" "<base gloss>"  %subsective semantics
    <stem> == SUFFIXATION.

LFT_AUGMENTATIVE:
    <> == LFT_CAT_PRESERVING
    <sem feature> == big
    <deriv affix> == ishch
    <declensional_class> == N_IV:<mor>.



LFT_DIMINUTIVE:
    <> == LFT_CAT_PRESERVING
    <sem feature> == small
    <deriv affix> == _k
    <declensional_class> == N_II:<mor>.

LFT_HEAD_MARKING:
    <> == LFT_CAT_PRESERVING
    <mor> == "<deriv affix>" "<base mor>"
    <stem> == PREFIXATION.

LFT_NEG_ADJ:
    <> == LFT_HEAD_MARKING
    <deriv affix> == ne
    <sem feature> == not.

LFT_INTENSIFIER_ADJ:
    <> == LFT_HEAD_MARKING
    <deriv affix> == pre
    <sem feature> == extremely.

% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%
%                                                                           
%
%       4.  SAMPLE LEXICAL ENTRIES                                          
%
%                                                                           
%
% % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % % 
%

Chitat':
    <> == VERB
    <gloss> == read
    <root all> == chit
    <stem> == <root all> a
    <syn valence> == 2.    %overrides default at LEXEME

Chitatel':
    <> == LFT_PERSON
    <base> == "Chitat':<>".

Dobro:
    <> == NOUN



    <declensional_class> == N_IV:<mor>
    <gloss> == good_deed
    <root all> == dobr
    <index> == 2
    <mor pl> == undefined.

Dobrij:
    <> == LFT_QUAL_ADJ
    <base> == "Dobro:<>".

Predobrij:
    <> == LFT_INTENSIFIER_ADJ
    <base> == "Dobrij:<>".

Reshit':
    <> == VERB
    <gloss> == decide
    <root all> == resh
    <stem> == <root all>
    <syn case_assign> == instrumental.

Reshenijo:
    <> == LFT_NOMINALIZATION
    <base> == "Reshit':<>".
  %  <syn case_assign> == Reshit'.

Gramotnij:
    <> == ADJ
    <gloss> == literate
    <stem> == gramotn.

Negramotnij:
    <> == LFT_NEG_ADJ
    <base> == "Gramotnij:<>".
    
Rabota:
    <> == NOUN
    <declensional_class> == N_II:<mor>
    <gloss> == work
    <root all> == rabot.

Rabotka:
    <> == LFT_DIMINUTIVE
    <base> == "Rabota:<>".



Dom:
    <> == NOUN
    <declensional_class> == N_I:<mor>
    <gloss> == house
    <root all> == dom.

Dom'ischcho:
    <> == LFT_AUGMENTATIVE
    <base> == "Dom:<>".

    

    


